

4 praktikum – üldine hindamisteoreem, iml kasutamine SAS-is

Kõigepealt lõpetame 3. praktikumi. On jäänud leida veel $V(\hat{t})$ Poissoni disaini korral.

4.1 Üldine hindamisteoreem + LJV TGA

3. praktikumis leidsime hinnangu tunnuse *totinch* kogusummale (on kirjas ka Wiki Moodles). Saadud hinnangu dispersiooni hindasime Monte-Carlo meetodil (samuti on olemas 3. nädala Wikis).

Hindame dispersiooni üldise hindamisteoreemi abil.

Ülesanne 1. Kirjutada välja üldise hindamisteoreemi valemid hinnangu \hat{t} jaoks koos dispersiooni hinnanguga $\hat{V}(\hat{t})$. Viimase jaoks panna kirja ka valem maatrikskujul.

Ülesanne 2. Panna kirja LJV TGA jaoks järgmised arvkarakteristikud (vt. 3 loengu slaidid):

$$EI_i = \quad ; \check{\Delta}_{ii} = \quad ; \check{\Delta}_{ij} =$$

Kasutage valimit mahuga $n = 10$, mis on eelmine praktikum saadud LJV TGA abil. Kui pole alles, siis tuleks uuesti võtta.

Koostame programmi, mis võimaldab leida hinnangu *totinch* kogusummale ning selle hinnangu standardviga. Selleks kasutame iml-keelt, mis käsitleb vektoreid ja maatrikseid. Programmi kopeerimisel pdf-st tuleb muuta ülakomasid!

```
proc iml;
start;
nn=50; use valimLJVTGA;
read all var {'totinch'} into y;
read all var {'NumberHits'} into k;
n=nrow(y);
w=(nn/n)#k; /*# on skalaaride korrutamine ja * on maatrikskorrutamine*/
yw=y#w;
t_hinnang=sum(yw);
delta=j(n,n,-1/(n-1)); *loob n*n maatriksi elementidega -1/(n-1);
do i=1 to n; /*muudame peadiagonaali*/
delta[i,i]=(nn-1)/(nn-1+n);
end;
disp=t(yw)*delta*yw; /*t() on maatriksi transponeerimine*/
st_dev=sqrt(disp);
suht=st_dev/t_hinnang;
```

```
nimed={'Kogusumma hinnang' 'Standardviga' 'Suhteline viga'};
tulemus=t_hinnang||st_dev||suht;
print tulemus[colname=nimed];
finish;
run;
quit;
```

Tähtis! Võib tulla veateade, et ei ole võimalik võtta ruutjuurt negatiivsest dispersioonist! Programmis pole viga, vaid kuna tgemist on dispersiooni HINNANGUGA, siis võib tulla ka negatiivne väärtus. See on üldise hindamisteoreemi eripära! Kui tulemus on positiivne, siis kannu hinnangu t -le koos cv-ga Moodle 4.nädala Wikisse.

4.2 SYG teoreem + LJV TGA

Ülesanne 3. Kirjutada välja SYG teoreemi valemid hinnangu \hat{t} jaoks koos dispersiooni hinnanguga $\hat{V}(\hat{t})$.

SYG teoreemi järgse standardviga saab leida järgmise programmi abil (sisestada Wikisse):

```
proc iml;
start;
nn=50; use valimLJVTGA;
read all var {'totinch'} into y;
read all var {'NumberHits'} into k;
n=nrow(y);
ylaiend=y#nn/n;
t_hinnang=sum(ylaiend#k);
delta_ij=-1/(n-1);
summa=0;
do i=1 to n;
abi=0;
do j=1 to n;
abi=abi+k[i]*k[j]*delta_ij*(ylaiend[i]-ylaiend[j])**2;
end;
summa=summa+abi;
end;
disp=-1/2#summa;
st_dev=sqrt(disp);
suht=st_dev/t_hinnang;
nimed={'Kogusumma hinnang' 'Standardviga' 'Suhteline viga'};
tulemusSYG=t_hinnang||st_dev||suht;
print tulemusSYG[colname=nimed];
finish;
run;
quit;
```